
require 'plot numeric jpeg viewmat'

NB. ************************************************************
NB. Principal Analysis
NB. common util 

   mean=:+/ % #             NB. mean

   mp=: +/ . *              NB. inner products

NB. Correlation matrix

   dev=: -"1 mean            NB. residue for multiple("1)

   ss=: +/@:*:@dev           NB. square sum of residue

   sd=: %:@(ss % #)          NB. standard deviation

   stand=: dev % "1 sd     NB. standardization

   

   cov=: (|: mp ]) % #      NB. Valiance Covariance matrix 

   cor=: cov@stand          NB. Correlation atrix

NB.    principal=: (stand DAT) mp pick_evec corr DAT

   principal=: stand mp pick_evec@cor 

   reg=: %. 1&,.@]    NB. rinear regression

   reg1=: 3 : '({: "1 y) %. 1&,. {."1 y'

plot_reg=: 3 : 0
NB. Usage: u STYLE //2 principals
DAT0=. 2{."1   y
REG=. reg1 DAT0 
DAT1=. REG&p. {. |: DAT0
pd 'reset'
pd 'type marker'
pd { |: DAT0
pd 'color red'
pd 'type line'
pd {|: /:~({."1 DAT0),.DAT1
pd 'show'
)
NB. ------------------------------------
NB. sample data(G.Suzuki) propotion of Miss Universe 

STYLE=: 165 53 86 56 92 160 47 84 52 92 166 55 86 64 89 164 56 90 60 95 
STYLE=: STYLE,168 55 87 56 87 164 54 87 57 92 168 54 94 58 97 169 55 88 57 92 
STYLE=: STYLE,169 53 86 58 93 166 56 84 57 90 165 53 85 55 90 163 49 84 59 90
STYLE=: STYLE,164 52 87 58 90 167 53 86 59 88 169 58 89 60 90 169 51 84 60 90
STYLE=: STYLE,166 50 86 59 87 168 53 88 60 88 165 54 88 62 90 167 50 88 58 89
STYLE=: STYLE,170 55 88 60 90 168 57 84 62 92 168 56 85 62 94 163 52 83 60 88
STYLE=: 24 5 $ STYLE

NB. ****************************************************
NB. -----------------
NB. find Eigenvalue and Eigenvector
NB. Levierre Faddeev Method
NB. -------------------------------
tr=: (<0 1)&|:               NB. trace
NB. umatrix=: (=/~)@i.@#

char_lf=: 3 : 0
ANS=.TR_SUM=. +/ tr MAT=. y NB. sum of trace



UMAT=. =/~ i. # y
for_LF. i.<: # y do.
MAT=. y +/ . * MAT - UMAT *  TR_SUM
TR_SUM=. (% 2+LF)* +/  tr MAT
ANS=. ANS,TR_SUM
end.
(p. POL), (<POL=.(-|. ANS),1)
)

char_lf_evec=: 3 : 0
EIGEN=. {@> ; 1{ char_lf y
EIGEN2=. {@> L:0 EIGEN ^/ L:0 |. i. # EIGEN
ADJMAT=. char_lf_evec_sub y
ANS=. <''
for_LF. i. # y do.
TMP=. +/> ( > LF{ EIGEN2) * L:0 ADJMAT
ANS=. ANS,<TMP

 end.
EIGEN,:}. ANS
)

char_lf_evec_sub=: 3 : 0
NB. modifyed char_lf
TR_SUM=. +/ tr MAT=. y NB. sum of trace
ANS=. <UMAT=. =/~ i. # y
for_LF. i.<: # y do.
MAT=. y +/ . * TMP=. MAT - UMAT *  TR_SUM
TR_SUM=. (% 2+LF)* +/  tr MAT
ANS=. ANS,<TMP
end.
)

norm0=: ] % [:%: [: +/ *:

pick_evec=: 3 : 0
TMP=.,./}. char_lf_evec y NB. jisuuno narashi
|: ;("1) norm0 L:0 ,.({@> i. # TMP){"1 L:0 TMP
)

NB.+++++++++++++++++++++++++++++++++++++++
NB. Cluster analysis
NB. Script is written in Jsoftware.com Essay/Dendrite 

 dist=:+/&.:*:@:-"1/~

 dist2=: %:@(+/)@:*:@:-"1/~   NB. same dist

NB. Usage:  7 mstc dat1
mstc=: 4 : 0

 z=. 0 2$f=. i.k=. x[w=. 0$~0,x

 for_e. y do.

  if. ~:/j=. |.^:(>/) e{f do.

   z=. z,e

   if. 1=k=.<:k do. z;w,x#0 return. end.

   w=.w,f=. ({.j) (f I.@:= {:j)} f

  end.

 end.

 assert. 0 [ 'graph is not connected'
)



  boxclust=: (</. i.@#)"1
NB. USAGE:   boxclust 1{::7 mstc dat1

   od2=: 2&# #: i.@*:   NB. 2 coord odometer

   edgesort=: ((] /: ({~ <"1)) (#~ </"1)) od2@#
NB. USAGE:  |:edgesort dist2 dat2
NB. Usage:D=: boxclust 1{:: (# mstc edgesort) dists dat2

nok=: 3 : 0

  z=. (i.{:$y),}:y

  for_i. i.&.<:#y do. z=. z /:"1 i{z end.
)

trimbox=: {.@(];.0~ _2 _2&({.!.1)@(-&2)@{:@$)@":
iceview=: (# {. }.) ([: trimbox </.)"1 {.
linkage=: {~ <"1
icicle=: {.@] , <@:iceview@nok@(1&{::)@] , (<@,.@linkage 0&{::)
NB. Usage:   (icicle (# mstc edgesort)) dist2 4|. dat2

silnok=: 3 : 'nok 1{:: (# mstc edgesort) dist y'
silview=: viewmat@silnok
NB. Usage: silview dat2

  

 nsf=: #~ _2&(+./\ ~:)@,

NB. ---------DATABlock---------------

NB. sample data (T.Suga)
DAT0=:1 165  50  110
DAT0=: DAT0,2 165  55  97
DAT0=: DAT0,3 160 52 108
DAT0=:DAT0,4 175 65  96
DAT0=: DAT0,5 175 68 105
DAT0=: DAT0,6 165 70  96
DAT0=: 6 4 $ DAT0

NB. Showcase dentride

  dat1=: _2]\1 3 5 6 3 4 3 6 4 6 3 5 2 5 1 4 0 1 2 3 0 2 0 3

 NB.   7 mstc e

dat2=: sharma=: _2]\5 5  6 6  15 14  16 15  25 20  30 19

NB. Suzuki 7 type of dogs upper-jaw
dat3=: 9.7 21.0 19.4 7.7 32.0 36.5
dat3=: dat3,8.1 16.7 18.3 7.0 30.3 32.9
dat3=: dat3,13.5 27.3 26.8 10.6 41.9 48.1
dat3=: dat3,11.5 24.3 24.5 9.3 40.0 44.6
dat3=: dat3,10.7 23.5 21.4 8.5 28.8 37.6
dat3=: dat3,9.6 22.6 21.1 8.3 34.4 43.1
dat3=: dat3,10.3 22.1 22.1 8.1 32.3 35.0
dat3=: 7 6 $ dat3
NB. ------------------------------

NB. Arima&Ishimura p80
dat4=: 22.9 24.9 19.3 22.0 28.6 42.6 41.3
dat4=: dat4,.13.7 16.2 11.3 10.4 24.9 26.5 20.3


